# $URL: http://mingw-lib.googlecode.com/svn/trunk/c-drive/mingw-lib/library/t_core/t_core.pro $
# $Rev: 859 $
# $Author: akio.miyoshi $
# $Date:: 2013-07-20 13:47:47 +0000#$
QT += network xml
TARGET = t_core
TEMPLATE = lib
CONFIG += staticlib

DEFINES += T_CORE_LIBRARY
SOURCES += \ 
    t_core/t_core.cpp \
    t_core/t_thread.cpp \
    t_core/nbr.cpp \
    t_core/t_lock.cpp
HEADERS += \
    t_core/t_thread.h \
    t_core/t_time.h \
    t_core/t_memory.h \
    t_core/t_lock.h \
    t_core/t_global.h \
    t_core/t_data.h \
    t_core/nbr.h \
    t_core.h
OTHER_FILES += \
    t_core.prf

#[export]
build_pass:CONFIG(debug, debug|release) {
    TARGET = $$join(TARGET,,,_debug)
}
CONFIG(release, debug|release) {
    target.path = C:/usr/local/release
    INSTALLS += target
} else {
    target.path = C:/usr/local/debug
    INSTALLS += target
}
includes1.path = C:/usr/local/include/t_core
includes1.files = *.h
INSTALLS += includes1
includes2.path = C:/usr/local/include/t_core/t_core
includes2.files = t_core/*.h
INSTALLS += includes2

features.path = $$[QMAKE_MKSPECS]/features
features.files = t_core.prf
INSTALLS += features
